Towards Stirling's Approximation

Ruben Gamboa

Department of EECS
University of Wyoming
ruben@uwyo.edu

ACL2 Workshop 2025
Austin, TX

Stirling's Approximation

Note: This talk is about **current, ongoing** work.

Stirling's Approximation

Note: This talk is about **current, ongoing** work.

$$n! \sim \sqrt{2\pi}\,\sqrt{n}\,n^n\,e^{-n}$$

Stirling's Approximation

Note: This talk is about **current, ongoing** work.

$$n! \sim \sqrt{2\pi}\,\sqrt{n}\,n^n\,e^{-n}$$

Stirling's contribution: The constant factor $\sqrt{2\pi}$

Standard mathematical proof is to start with

$$n! \sim K\,\sqrt{n}\,n^n\,e^{-n}$$

and then show $K = \sqrt{2\pi}$

Stirling's Approximation

Note: This talk is about **current, ongoing** work.

$$n! \sim \sqrt{2\pi}\,\sqrt{n}\,n^n\,e^{-n}$$

Stirling's contribution: The constant factor $\sqrt{2\pi}$

Standard mathematical proof is to start with

$$n! \sim K\,\sqrt{n}\,n^n\,e^{-n}$$

and then show $K = \sqrt{2\pi}$

One way to show the last part is to use Wallis's Product Formula for π (what this talk is about)

Wallis's Product Formula

$$\frac{\pi}{2} = \prod_{n=1}^{\infty} \left(\frac{2n}{2n-1} \cdot \frac{2n}{2n+1} \right) = \left(\frac{2}{1} \cdot \frac{2}{3} \right) \left(\frac{4}{3} \cdot \frac{4}{5} \right) \left(\frac{6}{5} \cdot \frac{6}{7} \right) \cdots .$$

Wallis's Product Formula

$$\frac{\pi}{2} = \prod_{n=1}^{\infty} \left(\frac{2n}{2n-1} \cdot \frac{2n}{2n+1} \right) = \left(\frac{2}{1} \cdot \frac{2}{3} \right) \left(\frac{4}{3} \cdot \frac{4}{5} \right) \left(\frac{6}{5} \cdot \frac{6}{7} \right) \cdots .$$

What we did:

1. Necessary improvements to the theory of calculus in ACL2(r)
2. Results and calculations for $W(n) \triangleq \int_0^{\pi} \sin^n x \, dx$
3. Reasoning about $W(n)$

1. Improvements to Calculus in ACL2(r)

- The theory of calculus (continuity, differentiability, integration) relies on non-standard analysis, and uses `encapsulate` to introduce (continuous, etc.) functions of a single variable
- E.g., “Let $f(x)$ be a continuous function. Then…”

1. Improvements to Calculus in ACL2(r)

- The theory of calculus (continuity, differentiability, integration) relies on non-standard analysis, and uses `encapsulate` to introduce (continuous, etc.) functions of a single variable
- E.g., "Let $f(x)$ be a continuous function. Then…"

- In calculus textbooks, we generalize this to multivariabe functions by "holding other variables constant"
- In ACL2, this is accomplished by using pseudo-lambda expressions with functional instantiation:

```
(:functional-instance ftc-2
                      (f (lambda (x) (g x y)))
```

- But this is not allowed in ACL2(r) — with good reason!

1. Improvements to Calculus in ACL2(r)

```
(:functional-instance ftc-2
                      (f (lambda (x) (g x y)))
```

- But this is not allowed in ACL2(r) — with good reason!

- This is a longstanding issue
- We've done ad hoc solutions in the past
  - Custom version of needed theorems for Taylor Series
  - Custom (and complex-valued) version of needed theorems for Fundamental Theorem of Algebra
- Now weve adopted the “context” solution from the FTA proof in ACL2(r)

1. More Improvements to Calculus in ACL2(r)

- Integration by Parts

$$\int_a^b u\,dv = uv|_a^b - \int_a^b v\,du$$

- This follows “trivially” from the chain rule (previously done in ACL2(r))

2. The Wallis Integral

$$W(n) \triangleq \int_0^{\pi} \sin^n x \, dx$$

- Using FTC-2:
  - $W(0) = \int_0^{\pi} \sin^0 x \, dx = \pi$
  - $W(1) = \int_0^{\pi} \sin^1 x \, dx = 2$

2. The Wallis Integral

$$W(n) \triangleq \int_0^{\pi} \sin^n x \, dx$$

- Using FTC-2:
  - $W(0) = \int_0^{\pi} \sin^0 x \, dx = \pi$
  - $W(1) = \int_0^{\pi} \sin^1 x \, dx = 2$
- Using `integral-rifn-small-<=-integral-rifn-big`:
  - $W(n+1) \leq W(n)$

2. The Wallis Integral

$$W(n) \triangleq \int_0^{\pi} \sin^n x \, dx$$

- Using FTC-2:
  - $W(0) = \int_0^{\pi} \sin^0 x \, dx = \pi$
  - $W(1) = \int_0^{\pi} \sin^1 x \, dx = 2$
- Using `integral-rifn-small-<=-integral-rifn-big`:
  - $W(n+1) \leq W(n)$
- Using Integration by Parts:
  - $W(n) = \frac{n-1}{n} W(n-2)$

2. The Wallis Integral

$$
\begin{aligned}
\int_0^{\pi} \sin^n x \, dx &= \int_0^{\pi} \sin^{n-1} x \, \sin x \, dx \\
&= (\sin^{n-1} x)(-\cos x)\Big|_0^{\pi} - \int_0^{\pi} (-\cos x)\left((n-1)\sin^{n-2} x \, \cos x\right) dx \\
&= (n-1)\int_0^{\pi} \sin^{n-2} x \, \cos^2 x \, dx \\
&= (n-1)\int_0^{\pi} \sin^{n-2} x \left(1 - \sin^2 x\right) dx \\
&= (n-1)\int_0^{\pi} \sin^{n-2} x \, dx - (n-1)\int_0^{\pi} \sin^n x \, dx
\end{aligned}
$$

2. The Wallis Integral

So,

$$n \int_0^{\pi} \sin^n x \, dx = (n-1) \int_0^{\pi} \sin^{n-2} x \, dx$$

and

$$\int_0^{\pi} \sin^n x \, dx = \frac{n-1}{n} \int_0^{\pi} \sin^{n-2} x \, dx$$

3. Reasoning about $W(n)$

```
(defun wallis-expansion (n)
   (if (zp n)
       (acl2-pi)
       (if (equal n 1)
           2
           (* (/ (- n 1) n)
              (wallis-expansion (- n 2))))))

 (defun wallis-factor (n)
   (if (and (integerp n)
             (<= 2 n))
       (* (/ (- n 1) n)
          (wallis-factor (- n 2)))
       1))

;;; (wallis-factor 10) => (*  9/10 7/8 5/6 3/4 1/2)
;;; (wallis-factor 11) => (* 10/11 8/9 6/7 4/5 2/3)
```

3. Reasoning about $W(n)$

```
(defthm wallis-expansion-for-evens
    (implies (and (natp n) (evenp n))
             (equal (wallis-expansion n)
                    (* (acl2-pi)
                       (wallis-factor n)))))

(defthm wallis-expansion-for-odds
    (implies (and (natp n) (not (evenp n)))
             (equal (wallis-expansion n)
                    (* 2
                       (wallis-factor n)))))

(defthmd wallis-as-wallis-expansion
    (implies (natp n)
             (equal (wallis n)
                    (wallis-expansion n))))
```

3. Reasoning about $W(n)$

```
(defthmd wallis-triple-bounds
    (implies (and (natp n) (<= 2 n))
             (and (<= (wallis (1+ n))
                      (wallis n))
                  (<= (wallis n)
                      (wallis (1- n))))))

(defthmd wallis-squeeze-bounds
    (implies (and (natp n) (<= 2 n))
             (and (<= 1
                      (/ (wallis n)
                         (wallis (1+ n))))
                  (<= (/ (wallis n)
                         (wallis (1+ n)))
                      (/ (1+ n) n)))))
```

3. Reasoning about $W(n)$

```
(defthmd wallis-quotient-asymptotic
      (implies (and (natp n) (i-large n))
               (equal (standard-part (/ (wallis n)
                                        (wallis (1+ n))))
                      1)))

(defthmd wallis-quotient-asymptotic-2
      (implies (and (natp n) (i-large n))
               (equal (standard-part (/ (wallis n)
                                        (wallis (1- n))))
                      1)))
```

3. Reasoning about $W(n)$

```
(defthm wallis-product-lemma
    (implies (and (natp n)
                  (not (evenp n))
                  (i-large n))
             (equal (standard-part (/ (wallis-factor n)
                                      (wallis-factor (1- n))))
                    (/ (acl2-pi) 2))))
```

3. Reasoning about $W(n)$

```
(defun wallis-product (n)
  (if (zp n)
      1
      (* (/ (* 2 n) (1- (* 2 n)))
         (/ (* 2 n) (1+ (* 2 n)))
         (wallis-product (1- n)))))

 (defthm wallis-product-lemma-1
     (implies (natp n)
              (equal (wallis-product n)
                     (/ (wallis-factor (1+ (* 2 n)))
                        (wallis-factor (* 2 n))))))
```

3. Reasoning about $W(n)$

```
(defthm wallis-product-convergence
    (implies (and (natp n)
                  (i-large n))
             (equal (standard-part (wallis-product n))
                    (/ (acl2-pi) 2))))
```

Lagniappe

```
ACL2(r) !>(* 2 (wallis-product 10))
137438953472/44801898141
```

Lagniappe

```
(defun df-wallis-product (n)
  (if (zp n)
      (to-df 1)
      (df* (df/ (df* 2 (to-df n))
                (df- 1 (df* 2 (to-df n))))
           (df/ (df* 2 (to-df n))
                (df+ 1 (df* 2 (to-df n))))
           (df-wallis-product (1- n)))))
```

Lagniappe

```
ACL2(r) !>(df* 2 (df-wallis-product 10))
#d3.067703806643497
```

Lagniappe

```
ACL2(r) !>(df* 2 (df-wallis-product 10))
#d3.067703806643497

ACL2(r) !>(df* 2 (df-wallis-product 100))
#d3.133787490628159
```

Lagniappe

```
ACL2(r) !>(df* 2 (df-wallis-product 10))
#d3.067703806643497

ACL2(r) !>(df* 2 (df-wallis-product 100))
#d3.133787490628159

ACL2(r) !>(df* 2 (df-wallis-product 1000))
#d3.1408077460303865
```

Lagniappe

```
ACL2(r) !>(df* 2 (df-wallis-product 10))
#d3.067703806643497

ACL2(r) !>(df* 2 (df-wallis-product 100))
#d3.133787490628159

ACL2(r) !>(df* 2 (df-wallis-product 1000))
#d3.140807746030865

ACL2(r) !>(df* 2 (df-wallis-product 10000))
#d3.1415141186818567
```